Rozhledy matematicko-fyzikální

Stanislav Trávníček

Co s prošlým kalendářem

Rozhledy matematicko-fyzikální, Vol. 88 (2013), No. 1, 16–21

Persistent URL: `http://dml.cz/dmlcz/146507`

Co s prošlým kalendářem

Stanislav Trávníček, PřF UP, Olomouc

Abstract. The article deals with the following problem: Given a particular year, we look for other years with the same calendar, which means the years where the 1st of January is the same day of the week as in the original year. We can also look for the years which have the same calendar as the original year only after the 1st of March.

Jednou se panu Koumalovi stalo, že si již na podzim koupil kalendář na příští rok, ale někam ho založil, takže si musel koupit nový. Ten původní pak objevil v krabici se starými fotografiemi, ale až za rok, nedlouho před koncem jeho platnosti. Takže ho zcela nepoužitý vyhodil.

Pak se však zamyslel, trochu vážně a trochu nevážně, že ho vlastně vyhazovat nemusel, ale mohl si ho uschovat, až se v některém z budoucích roků bude zase hodit. Kdy to bude? Jak to zjistit? Tento problém stále nechtěl jeho mysl opustit, a tak si pan Koumal řekl, že se na to podívá blíže.

Původní letopočet si označil $R0$ a příští rok, kdy by kalendář byl „shodný“ s kalendářem z roku $R0$, označil $R1$. Jeho problém tedy zněl: Jak k danému roku $R0$ určíme rok $R1$? I když je počítačový amatér, tak se rozhodl, že na ten problém sestaví počítačový program. Napřed že však samozřejmě musí celou věc důkladně prozkoumat.

Hned na začátku ho napadlo, že u přestupných roků to bude jinak, že pokud je R0 přestupný, musí být $R1$ také přestupný, a navíc 1. leden v obou těchto letech musí být stejný den v týdnu, třeba úterý. Jistě není důležité, je-li to úterý nebo třeba čtvrtek, ale důležité je, aby to byl týž den v týdnu. Pan Koumal si dny v týdnu označil čísly (indexy) 0 až 6 a věděl, že nebude potřeba přidělovat tato čísla konkrétním názvům dnů v týdnu. Stanovil si, že rok R0 prostě začíná dnem 0 a že nás zajímají roky, které také začínají dnem 0. Dokonce si pomocí tohoto indexu dne definoval, které roky (a jejich kalendáře) bude považovat při řešení svého problému za shodné: Roky $R0$ a $R1$ bude považovat za shodné, pokud jsou oba přestupné, nebo oba nepřestupné (říkal jim „obyčejné“) a u obou je 1. leden týmž dnem 0 v týdnu.

Jako první si pan Koumal zjišťoval, jak se změní index prvního ledna od roku $R0$ do roku $R0+4$. Týden má 7 dní, a tak bylo jasné, že tu hraje důležitou úlohu dělení sedmi, tedy zbytky při dělení sedmi. Zavedl si proto označení, že $Zb(n)$ bude znamenat zbytek při dělení čísla n sedmi, takže pro přestupný rok to dává $Zb(366) = 2$; je-li tedy $R0$ přestupný rok začínající dnem 0, je 1. leden roku $R0 + 1$ dnem s indexem 2. Pro obyčejný rok $R0$ je $Zb(365) = 1$, takže 1. leden roku $R0 + 1$ je dnem s indexem 1. Z toho pak dostal, že za 4 roky se index 1. ledna zvětší o 5 $(= 2+1+1+1)$, takže když 1. leden v roce R0 má index 0, má 1. leden o 4 roky později, v roce $R0 + 4$, index 5.

Případ, že $R0$ je přestupný rok, vyřídil pak Koumal velmi brzo; sestavil si posloupnost indexů 1. ledna v přestupných rocích:

$$0,\ 5,\ Zb(5+5)=3,\ Zb(3+5)=1,\ Zb(1+5)=6,$$
$$Zb(6+5)=4,\ Zb(4+5)=2,\ Zb(2+5)=0;$$

kalendář shodný s rokem $R0$ bude až za 28 roků. Také si uvědomil, že interval 4 mezi přestupnými roky je s počtem 7 dnů v týdnu nesoudělný, takže každé 4 roky bude mít index 1. ledna jinou hodnotu a protože možností je 7, nastane 0 až po sedmi čtyřletích, tj. za 28 roků.

Pak začal uvažovat o obyčejných rocích, ale žádné podobné jednoduché pravidlo hned neobjevil, a tu ho problém teprve začal opravdu zajímat. Uvědomil si, že jsou letopočty tří druhů, $4k+1$, $4k+2$, $4k+3$, kde $4k$ je přestupný rok. Aby do problému trochu blíže pronikl, začal si, počínaje rokem typu $4k+1$, postupně vypisovat indexy 1. ledna několika následujících let; za index 1. ledna roku $4k+1$ vzal 0; dochází postupně k těmto změnám:

$$0 \to 1, \quad 1 \to 2, \quad 2 \to \mathbf{3}, \quad \mathbf{3} \to 5 \quad \text{atd.}$$

To dává posloupnost

$$0,\ 1,\ 2,\ \mathbf{3},\ 5,\ 6,\ 0,\ \mathbf{1},\ 3,\ 4,\ 5,\ \mathbf{6},\ 1,\ 2,\ \ldots \tag{1}$$

(po dnu 6 přichází den $Zb(6+1) = 0$; tučně jsou uvedeny kódy 1. ledna v přestupných rocích). „No vida," řekl si pan Koumal, „první shodný kalendář dostaneme už za 6 let." Podobně zkusil vypsat posloupnost indexů i pro případ, že výchozí rok je typu $4k+2$:

$$0,\ 1,\ \mathbf{2},\ 4,\ 5,\ 6,\ \mathbf{0},\ 2,\ 3,\ 4,\ \mathbf{5},\ 0,\ 1,\ 2,\ \ldots$$

Opět to bylo 6 let. Je to tedy to hledané pravidlo? Ale tu si pan Koumal všimnul, že ten nejbližší rok s nulou je přestupný, takže shoda kalendářů nenastala, ta první bude až po 11 letech. Protože mu výsledky 6 a 11 připadly dost nesourodé, tak prošetřil ještě rok typu $4k+3$:

$$0,\ \mathbf{1},\ 3,\ 4,\ 5,\ \mathbf{6},\ 1,\ 2,\ 3,\ \mathbf{4},\ 6,\ 0,\ \ldots$$

Zase 11. A kdy budou další shody? Bylo mu jasné, že stejně jako u přestupných let se i zde musí situace vždy po 28 letech opakovat. Ale kolikrát za tu dobu shoda nastane? Pustil se do toho a počítal a počítal a nakonec zjistil, že v rámci těch 28 let dostáváme 2 shodné roky, a to po 6 a pak po 11 letech, nebo po 11 a pak po 6 letech, nebo po 11 a pak opět po 11 letech.

Už se chtěl dát do programování, když si vzpomněl, že jednou počátkem března šel kolem papírnictví a za výlohou viděl nabídku kalendářů s velkou slevou. To by znamenalo, že ti ještě šetrnější by se v lednu a v únoru bez kalendáře třeba nějak obešli, a pak si koupí prošlý kalendář z minulých let, pokud se bude s daným rokem shodovat aspoň od 1. března.

Zamyslel se tedy ještě nad tímto novým problémem. Nejprve si uvědomil, že když se kalendáře ve dvou přestupných letech mají shodovat od 1. března, že se fakticky shodují už od 1. ledna, takže tu nic nového nedostaneme. Podobně platí i pro obyčejné roky, že když se jejich kalendáře shodují od 1. března, že se také shodují už od 1. ledna, takže ani zde nepřibude další možnost uplatnění starého kalendáře. Zbývá případ, že jeden rok je přestupný, druhý obyčejný a 1. březen je v obou případech stejný den v týdnu. A co zjistil dále:

Leden s únorem mají v obyčejném roce celkem

$$31+28=59$$

dnů, takže má-li 1. leden index dne v týdnu 0, má 1. březen téhož roku index $Zb(59)=3$; v přestupném roce je součet dnů v lednu a únoru

$$31+29=60, \quad Zb(60)=4,$$

takže vzhledem k 1. lednu jde o posun indexu o 4. To znamená, že posloupnost indexů 1. března můžeme odvozovat přímo z posloupnosti novoročních indexů. Pan Koumal si zapsal posloupnost (1) příslušnou k roku $R0 = 4k + 1$ týkající se 1. ledna a pod ni umístil posloupnost týkající se 1. března.

1. leden: 0, 1, 2, **3**, 5, 6, 0, **1**, 3, 4, 5, **6**, 1, 2, …

1. březen: 3, 4, 5, **0**, 1, 2, 3, **5**, 6, 0, 1, **3**, 4, 5, …

Ve druhém řádku je vidět, že v roce R0 má 1. březen index 3 a stejnou trojku najdeme o 6 let později. Ale pozor. Při porovnání s prvním řádkem viděl, že se oba kalendáře zcela shodují již od 1. ledna (jde o dva obyčejné roky). Další trojku najdeme 11. rok (který je přestupný) po nepřestupném roce $R0$, takže skutečně se tu dostává případ částečné shody, tedy shody kalendářů až od 1. března. Podobně by se mohly vyšetřovat i další případy a v delších posloupnostech, ale tu si pan Koumal řekl dost!, zbytek ať prozkoumá počítač.

Předně uvážil, že nebude hledat jen nejbližší rok $R1$ shody s počátečním rokem $R0$, ale vyhledá všechny shody v intervalu, který se do programu zadá. Zjišťování roků shodných kalendářů tedy prodlouží od zadaného počátečního letopočtu, který v programu nazve `Rok0`, až po zadaný koncový letopočet `RokZ`. Algoritmus výpočtu ponechal týž jako při „ručním“ výpočtu, tj. aby se zkoumal jeden rok za druhým a průběžně se oznamovaly shody.

V programu použil pan Koumal tato označení:

- `K` … udává, jakého typu je zpracovávaný Rok, tj. kolikátý rok je to po přestupném roce
- `NR0` … je 0, pokud je zadaný rok přestupný, a 1, pokud je obyčejný
- `NR` … je 0, pokud je zpracovávaný rok přestupný, a 1, pokud je obyčejný
- `LK` … index 1. ledna běžného roku (v roce `Rok0` je roven 0)
- `BK0` a `BK` … indexy 1. března v počátečním a v běžném roce

Sestavil tento jednoduchý program, kde jediným nefunkčním efektem je počáteční vymazání obrazovky:

```
program Kalendar;
uses Crt;
var
  Rok0, Rok, RokZ, NR0, NR, LK, BK0, BK, K: Integer;
begin
  ClrScr;
  Write('Máme kalendář z roku: ');
```

```
  ReadLn(Rok0);
  Write('Chceme testovat jeho použití do roku: ');
  ReadLn(RokZ);
  WriteLn('Daný kalendář se hodí na roky:');
  if Rok0 mod 4 = 0 then
  begin
       NR0 := 0; BK0 := 4
  end else
  begin
       NR0 := 1; BK0 := 3
  end;
  Rok := Rok0; LK := 0;
  repeat
       Inc(Rok);
       K := Rok mod 4;
       case K of
       0: begin LK := (LK + 1) mod 7;
            BK := (LK + 4) mod 7; NR := 0 end;
       1: begin LK := (LK + 2) mod 7;
            BK := (LK + 3) mod 7; NR := 1 end;
       2, 3: begin LK := (LK + 1) mod 7;
            BK := (LK + 3) mod 7; NR := 1 end
       end;
       if (LK = 0) and (NR = NR0) then
            WriteLn(Rok, ' od 1. ledna');
       if (BK = BK0) and (NR + NR0 = 1) then
            WriteLn(Rok, ' od 1. března')
  until Rok >= RokZ;
  ReadLn
end.
```

Program fungoval k plné autorově spokojenosti, i když si od samého počátku a po celou dobu, kdy program připravoval, byl vědom jeho slabin. Například, že užití programu je omezeno tím, že interval zjišťování shodnosti nesmí obsahovat rok 2100, který se trochu tváří jako přestupný, ale přestupný není. Že nesmí překročit ani rok 2200, to už ho tak nebolelo. A pak samozřejmě neřešil umístění velikonočních svátků s tím, že nový uživatel starého kalendáře by si musel tyto pohyblivé svátky v kalendáři prostě překreslit.

A také ho napadlo, že kdyby se v některém roce objevily za výlohou papírnictví staré kalendáře s nápisem „Kupte kalendáře z roku…, budou platit i v příštím roce“, že by se určitě našli kupci se smyslem pro recesi.

Pan Koumal se tímto problémem s kalendáři už dál zabývat nebude, do komplikací s rokem 2100 se mu nechce, protože tento letopočet bude až zadlouho. Pokud byste však vy chtěli platnost programu rozšířit i přes rok 2100, čeká vás docela pěkné programátorské cvičení.

(Autorkou ilustrace je Mgr. Jaroslava Čermáková.)

* * * * *

K článku V. Dlaba: Srdce trojúhelníku (str. 4–9):

Napoleonův trojúhelník

Vně nad stranami trojúhelníku ABC jsou sestrojeny rovnostranné trojúhelníky ABK, BCL a CAM. Středy těchto trojúhelníků (těžiště) jsou vrcholy rovnostranného trojúhelníku XYZ, který se nazývá Napoleonův trojúhelník trojúhelníku ABC (viz obrázek).

(pozn. redakce)